Appendix A

A.1 List of the Useful SPARQL Queries Overcoming the Required Objectives and Covering the Necessary Criteria

Query 1: Retrieve the algorithm of a given ML model

```
SELECT ?algo_name
WHERE
    {
        ?ml rdf:type sml:MachineLearningModel .
        ?ml sml:hasAlgorithm ?algo .
        ?algo sml:AlgorithmName ?algo_name .
        ?ml sml:smlID "1"^^xsd:string
    }
```

Query 2: Describe the training data set of a given ML model

```
SELECT ?dataitem ?value ?featname
WHERE
    {
        ?ml rdf:type sml:MachineLearningModel .
        ?ml sml:smlID "1"^^xsd:string
        {
            ?ml sml:hasTrainingDataSet ?tds .
            ?tds sml:isComposedOf ?dataitem .
             ?dataitem sml:dataitemValue ?value
        }
        UNION
        {
            ?ml sml:hasTrainingDataSet ?tds .
            ?tds sml:ReliesOn ?feature .
            ?feature sml:featureName ?featname
        }
    }
```

Query 3: Describe the testing data set of a given ML model

```
SELECT ?dataitem ?value ?featname
WHERE
    {
        ?ml rdf:type sml:MachineLearningModel .
        ?ml sml:smlID "1"^^xsd:string
        {
            ?ml sml:hasTestingDataSet ?tds .
            ?tds sml:isComposedOf ?dataitem .
            ?dataitem sml:dataitemValue ?value
        }
        UNION
        {?ml sml:hasTestingDataSet ?tds .
        ?tds sml:ReliesOn ?feature .
        ?feature sml:featureName ?featname}
    }
```

Query 4: Retrieve the performance of a given ML model (the scores and the metrics used to calculate the evaluation applied to the testing data)

```
SELECT ?metric ?scorevalue
WHERE
    {
        ?ml rdf:type sml:MachineLearningModel .
        ?ml sml:smlID "1"^^xsd:string .
        ?ml sml:hasTestingDataSet ?tds .
        ?tds sml:hasEvaluation ?eval .
        ?eval sml:hasScore ?score .
        ?score sml:isCalculatedBy ?metric .
        ?score sml:scoreValue ?scorevalue
    }
```

Query 5: Retrieve the metadata of a given ML model with those related to its algorithm / its training data set / its testing data set and to the evaluations applied to the testing data)

```
SELECT ?smldfeaturename  ?smldvalueval  ?algo ?algomdfeaturename
?algomdvalueval ?trds ?trdsmdfeaturename
?trdsmdvalueval ?tsds ?tsdsmdfeaturename ?tsdsmdvalueval
?eval ?evalmdfeaturename ?evalmdvalueval
WHERE
    {
        ?ml rdf:type sml:MachineLearningModel .
        ?ml sml:smlID "1"^^xsd:string
        {
            ?ml sml:hasAlgorithm ?algo .
            ?algo sml:hasAlgorithmMetaData ?algo_md .
            ?algo_md sml:hasFeature ?algomdfeature .
            ?algomdfeature sml:featureName ?algomdfeaturename .
            ?algo_md sml:hasValue ?algomdvalue .
            ?algomdvalue sml:valueVal ?algomdvalueval
        }
        UNION
        {
            ?ml sml:hasTrainingDataSet ?trds .
            ?trds sml:DSHasMetaData ?trds_md .
            ?trds_md sml:hasFeature ?trdsmdfeature .
            ?trdsmdfeature sml:featureName ?trdsmdfeaturename  .
            ?trds_md sml:hasValue ?trdsmdvalue .
            ?trdsmdvalue sml:valueVal ?trdsmdvalueval
        }
        UNION
        {
            ?ml sml:hasTestingDataSet ?tsds .
            ?tsds sml:DSHasMetaData ?tsds_md .
            ?tsds_md sml:hasFeature ?tsdsmdfeature .
            ?tsdsmdfeature sml:featureName ?tsdsmdfeaturename  .
            ?tsds_md sml:hasValue ?tsdsmdvalue .
            ?tsdsmdvalue sml:valueVal ?tsdsmdvalueval  .
            ?tsds sml:hasEvaluation ?eval .
            ?eval sml:hasEvaluationMetaData ?eval_md .
            ?eval_md sml:hasFeature ?evalmdfeature  .
            ?evalmdfeature sml:featureName ?evalmdfeaturename .
            ?eval_md sml:hasValue ?evalmdvalue .
            ?evalmdvalue sml:valueVal ?evalmdvalueval
        }
        UNION
        {
            ?ml sml:hasMetaData ?ml_md .
            ?ml_md sml:hasFeature ?smldfeature .
            ?smldfeature sml:featureName ?smldfeaturename .
            ?ml_md sml:hasValue ?smldvalue .
            ?smldvalue  sml:valueVal ?smldvalueval
        }
    }
```

Query 6: Find the application domain of each ML model and give a clear description of this domain

```
SELECT ?ml ?domain_name (GROUP_CONCAT(?featurename;SEPARATOR=",") AS
    ↪  ?features)
WHERE
    {
        ?ml rdf:type sml:MachineLearningModel .
        ?ml sml:hasApplicationDomain ?appdomain .
        ?appdomain sml:domainName ?domain_name .
        ?appdomain sml:isDescribedBy ?topic .
        ?topic sml:hasTopicFeature ?feature .
        ?feature sml:featureName ?featurename
    }
GROUP BY ?ml ?domain_name
```

Query 7: Describe the training data context of a given ML model

```
SELECT ?featurename ?operatorvalue ?featurevalue
WHERE
{
    ?ml rdf:type sml:MachineLearningModel .
    ?ml sml:smlID "1"^^xsd:string .
    ?ml sml:hasTrainingDataSet ?tds .
    ?tds sml:hasContext ?cont .
    ?cont sml:hasConstraint ?constr .
    ?constr sml:hasSourceOperand ?feature .
    ?feature sml:featureName ?featurename.
    ?constr sml:hasOperator ?op .
    ?op sml:operatorValue ?operatorvalue .
    ?constr sml:hasTargetOperand ?value .
    ?value sml:valueVal ?featurevalue
}
```